UNIVERSITY OF CALIFORNIA

LICK OBSERVATORY TECHNICAL REPORTS

No. 16

SERIAL DATA MULTIPLEXER--EL-331

T. P. Ricketts and L. B. Robinson

Santa Cruz, California
February 1976

INDEX

SERIAL DATA MULTIPLEXER (EL-331) FOR THE PDP 8/I

Introduction

A general purpose digital input-output system is currently in use with the PDP 8/I computer at the 120-inch telescope (a driver also exists at the Santa Cruz PDP 8/I). In order to avoid having to run a number of multiwire cables throughout the dome, with the attendant problems of noise and unreliability, a serial multiplexing scheme has been developed that will allow two way transmission of data using a single pair of coaxial cables. This system cannot at present be connected to the well-known CAMAC system. Hopefully, a CAMAC compatible interface will be added to the system in the future.

The Serial Data Multiplexer (EL-331) may be used by many devices to gain access to the computer. The multiplexer is connected to the computer by two coaxial cables; one for data and one for shift pulses. Devices connect to the multiplexer through a 17 pin connector (MS 3106A-20-29S). Each multiplexer chassis can handle up to 14 input "words" and can provide up to 4 output "words." Input or output words consist of 12 bits of data, plus pairs of levels and pulses in some cases. The computer can select up to 7 multiplexer units, thus providing a total of 98 input and 28 output words to various devices through the 120-inch dome. Appendix 'A' shows the interconnection of the present system of multiplexers.

Each multiplexer box has 14 input connectors and 4 output connectors. The inputs are labeled J1 through J14. J1-J12 also provide 2 pulses and 2 levels for control of the devices from the computer. Cables J15 to J18 are the output cables. They provide 12 bits of data each, plus 2 pulses. The pulses are paralleled from J5 to J8. See Appendix "B" or EL-331-2D$_1$ for the assignments of signals at the connectors. If the data to be transmitted exceeds 12 bits, several connectors may be connected to the same

device, one for each 12 bits of data. The outputs are driven by TTL registers which in turn are cleared or loaded by the computer. TTL logic levels are used throughout.

In order for the computer to select the proper device, two words are serialized and transmitted on the coax. One word is the desired data; the other word is a function code which selects the unit (from 0 to 7), and the cable (from 0 to 15), as well as determining the direction of data (whether input or output). See Appendix "C" or EL-331-2D$_1$, for bit assignments and device codes. The two 12 bit words (Function and Data) are transmitted serially as two 15 bit words (12 bits plus 3 error detection bits per word) from the computer to the multiplexer. The multiplexer decodes the function to select the proper cable, loads a register if required, and retransmits the function back to the computer as well as the desired data. The total time for the transmission to the multiplexer and back is about 20μsec. A flag is set in the computer at the end of the transmission and may be checked with a PDP8 Input-Output (IOT) Pulse. Another flag will be set if the error-check code bits do not match. This also may be checked by an IOT. If the multiplexer sees the wrong error-check code it will inhibit the desired function and simply retransmit the incorrect data (with error unchanged) back to the computer. See Pages 4 and 5, Appendix "D", or EL-331-2D$_2$ for the IOT assignments and examples of the programming for various devices.

The Automated Cassegrain Spectrograph is operated by a control box (or by the computer) through the multiplexer; using multiplexer unit #2 and cable jacks: J5, J6, J7, J15, J16, J17 on the "tub". The control box will be cycling continually to provide position readouts for the spectrograph. Thus, if the spectrograph is turned on, and the multiplexer lines are desired by the computer, a "look at me" pulse must be sent to the

spectrograph control. This will disable it after it finishes its current transmission. Since the spectrograph control must transmit three times for each control function, it could take 100μ sec before it gives control of the line to the computer. When the computer is finished with the line it must send a "look away" pulse to the spectrograph to reenable it.

MULTIPLEXER CODES

To read data or to transmit data, build a "Function" code by adding a Cable code (A) to a Crate Code (C). To send pulses or set levels, build a "Function" code by adding an Operation code (B) to a Crate code (C), and select the cable number by transmitting a Data Code (B2) as "Data." See examples on the next page. Note that a "1" on Data output is +4V, but 0V. is a "1" on data input.

A) To get data from

Cable No.; use	(Function) or "Cable" Code:
1	4010
2	2010
3	6010
4	1010
5	5010
6	3010
7	7010
8	410
9	4410
10	2410
11	6410
12	1410
13	5410
14	3410

To send data to

Cable No.	use	Cable Code
15		5020
16		3020
17		7020
18		420

Data in the accumulator will then be transmitted when PDP8 instruction 6357 is given.

B) To set levels, or send 10 microsec, +4 V pulses, use

(Function) or "Operation" Code	Operation:
4020	Pulse 1
2020	Pulse 2
6022 V=4; 6021 V=0	Level 3
1022 V=4; 1021 V=0	Level 4

V ≡ VOLTS

B2) Select

Cable:	by Data Code:	Cable:	by Data Code:
1	4000	7 and 17*	40*
2	2000	8 and 18*	20*
3	1000	9	10
4	400	10	4
5 and 15*	200*	11	2
6 and 16*	100*	12	1

*Note that pulses will appear on both cables simultaneously.

C)

Crate No.	Crate Code	Crate No.	Crate Code
1	200	4	40
2	100	5	240
3	300	6	140
		7	340

Both the function and data codes are transmitted when PDP8 instruction 6357 is executed, with the data code in the accumulator.

A) Transmit Pulse 1 to Cable 18, Cable 8 } MUX Unit 3 (Cass Tub)

```
TAD (4320)          / 4020 selects Pulse 1; 300 selects Unit 3.
6355                / load Function
CLA                 / clear Accumulator
TAD (20)            / 20 selects Cable 8
6357                / Load "Data" and transmit "Data" and Function
CLA
ETC                 / wait 20 micro sec, then test for errors.
```

B) Transmit Pulse 2 to Cables 6 and 16, Unit 3

```
TAD (2320)          / 2020 for pulse 2, 300 for unit 3
6355                / load Function
CLA
TAD (100)           / 100 for Cable 6
6357                / load data and transmit Data and Function
```

C) Clear (V=0) Level (3) on Cables 5 Unit 1 (Readout Room)

```
TAD (6221)          / 6020 selects level 3; 200 selects Unit 1; 1 to set 0V.
6355                / load function
CLA
TAD (200)           / 200 selects cable 5
6357                / load "Data", and transmit Function and Data
```

D) Set (V=+4) Level (4) on Cables 6, 12, Unit 4 (Coude or Spare)

```
TAD (1062)          / 1020 selects level 4; 40 selects Unit 4; 2 to set + 4V
6355                / load Function
CLA
TAD (101)           / 100 for Cable 6, 1 for Cable 12
6357                / load "Data", and transmit Data and Function
```

E) Read Data from Cable 8, Unit 5

```
TAD (650)           / 400=cable 8; 10=read; 240=unit 5
6355                / load function from accumulator
CLA
6357                / load no data and transmit
JMS Delay           / wait 20 micro sec
Error Test          / optional error test
6353                / read data to accumulator
```

F) Transmit Data to Cable 18, Unit 5

```
TAD (660)           / 400=cable 18; 240=unit 5; 20=send
6355                / load function from accumulator
CLA
TAD DATA
6357                / transmit data from accumulator, transmit function
Delay               / wait 20 micro sec
Error Test          etc.
```

MULTIPLEXER SEQUENCE



"Look at me" - turns off other MUX users
"Look away" - enables them again
When power first comes on, the remote units may be out of sync with transmitter. Up to 30 transmissions may be needed to synchronize. No other errors should occur unless hardware is bad, cable lengths are unequal (or improperly terminated), or the addressed multiplexer is turned off. The spectrograph error light will blink each time the computer takes control of the cable.

Example of Multiplexer Use:-- Load Sweeps. for Scanner.

(See Page 8)



STANDARD MUX DRIVER SUBROUTINE (See P13)



File 2 Tape 12R
June 3/74

```
•PALP
*OUT-S:SWEP
*
*IN-S:CON0,S:XCON,S:SWP1,S:SWP2
*
*
*
*
*OPT-T

ARG 1    0050

                /CON0
                XLIST
                PAUSE/
                /
                /XCON
                FIELD 1
                XLIST
                PAUSE/
                /
                /SWP1
                /X MCEN(D) OFFSETS CENTER BY + OR - D
                /S D=FMEMX(N,C,R,S,P,K);X MEMY(0,C,R--)--LOAD SWEEPS
                /MUX NO. IS 3;OR S IF NON-ZERO...P=1 TO BYPASS ERROR PRINT.
                /N=1 FOR NORMAL SWEEP..R=1 TO READ,0 TO SEND.
                /.K=1 TO BYPASS 4K SCANNER MEMORY.
                /C IS SWEEP CENTER
                /INPUT CABLE 7;OUTPUT CABLE 18.
                /
                /ASSUMES MUX CYCLE TIME<20 MICROSEC.
                /WILL HANG IN XMIT IF ARG7 SET AND MUX DISABLED.
                /
                DATA=ARG10
                FUNCP=ARG10H
                COUNT8=ARG9
                FUNC=ARG9H
                COUNT=ARG8H
                UNIT=ARG7H
                BYPASS=ARG6H
                /
                MUX=6350
                OKSKIP=1
                DUNSKP=2
                DAREAD=3
                FREAD=4
                FLOAD=5
                LAMOFF=6
                DLOAD=7
                GRAB=4
                /
                SYNSKP=6452
                FUNLOD=6453
                /
                *FNKB1+71
0735    1200    1200      /MEMX
0736    1201    1201      /MEMY
                *FNKB1+75
0741    0666    666       /MCEN
                *KB1+75
0235    6477    CENTER
                *KB1+71
```

```
0231   6515   MEMX
0232   6531   MEMY
              /
              *6477
6477   0000   CENTER,0
6500   1052     TAD ARG3
6501   3053     DCA ARG4
6502   1075     TAD NORMAL
6503   4744     JMS I SETUPX
6504   5677   CENGO,JMP I CENTER       /THIS ROUTINE SETS CENTER COUNTER ON
              /
6505   0300   UNIT0,300
6506   0200   UNIT1,200
6507   0100   UNIT2,100
6510   0300   UNIT3,300
6511   0040   UNIT4,40
6512   0240   UNIT5,240
6513   0140   UNIT6,140
6514   0340   UNIT7,340
6515   0000   MEMX,0
6516   1052     TAD ARG3
6517   7440     SZA
6520   7330     STL CLA RAR    /4000
6521   1345     TAD NORM0
6522   3075     DCA NORMAL
6523   1054     TAD ARG5
6524   7640     SZA CLA
6525   1340     TAD XREAD      /READ,NOT WRITE
6526   1342     TAD FUNCX
6527   4744     JMS I SETUPX
6530   5715     JMP I MEMX
              /
6531   0000   MEMY,0
6532   1054     TAD ARG5
6533   7640     SZA CLA
6534   1341     TAD YREAD      /READ,NOT WRITE
6535   1343   TAD FUNCY
6536   4744     JMS I SETUPX
6537   5731     JMP I MEMY
              /
6540   0400   XREAD,1000-400   /CHANGE WRITE TO READ
6541   0100   YREAD,200-100
6542   6630   FUNCX,6630
6543   7131   FUNCY,7131
6544   6644   SETUPX,SETUP
6545   3230   NORM0,3230
              /
6546   0000   SFUNL,0          /PUT FUNCTION IN SWEEP BOX
6547   7421   SETIN,MQL
6550   1777     TAD PORT18     /CABLE SELECT
6551   4776     JMS XMIT
6552   5347     JMP SETIN      /ERROR
6553   1775     TAD CABLE7
6554   7421     MQL
6555   7330     CLA STL RAR    /4000 FOR PULSE 1
6556   4776     JMS XMIT
6557   5347     JMP SETIN      /ERROR
6560   5746     JMP I SFUNL
              /
6561@  0000   DELAY,0 /LENGTH IN AC=12+N*4.5 MICRO SEC
```

```
6562  3031     DCA TEMPS0
6563  2031     ISZ TEMPS0
6564  5363     JMP .-1
6565  5761     JMP I DELAY
               /
6566  0000   SYNC,0
6567  6452     SYNSKP
6570  5367     JMP .-1
6571  6452     SYNSKP          /WAIT FOR 4K MEM. CYCLE
6572  7410     SKP
6573  5371     JMP .-2
6574  5766     JMP I SYNC
               /
6575  6771
6576  6600
6577  6766
               *6600
6600  0000   XMIT,0
6601  1037     TAD P20         /SEND
6602  1014     TAD UNIT
6603  6355     MUX FLOAD       /FUNCTION
6604  3016     DCA FUNC
6605  6354     MUX FREAD
6606  7041     CIA
6607  1016     TAD FUNC        /BE SURE CORRECT CODE LOADED
6610  7450     SNA
6611  5214     JMP GO
6612  4536     JMS I OCTPNX    /IF PDP 8 IO BAD PRINTS DELTA
6613  5231     JMP FAULT
6614  7501   GO,MQA
6615  6357     MUX DLOAD       /"DATA" AND TRANSMIT
6616  7344    @CLA CMA CLL RAL        /-2 IN AC FOR 21 USEC.
6617  4775     JMS I DELAYX
6620  6352     MUX DUNSKP
6621  5231     JMP FAULT
6622  6354     MUX FREAD
6623  7041     CIA
6624  1016     TAD FUNC
6625  6351     MUX OKSKIP
6626  7240     CLA CMA         /ERROR
6627  7650     SNA CLA
6630  5242     JMP OK
6631  1013   FAULT,TAD BYPASS
6632  1056     TAD ARG7
6633  7640     SZA CLA
6634  5600     JMP I XMIT      /BYPASS ERROR MESSAGE
6635  4422     JMS I MESAGX
6636  1525   TEXT /MU
6637  3077   X?
6640  0000   /
6641  5600     JMP I XMIT
6642  2200   OK,ISZ XMIT
6643  5600     JMP I XMIT
               PAUSE/
               /
               /SWP2
6644  0000   SETUP,0
6645  3017     DCA FUNCP
6646  6002     IOF
6647  6354     MUX GRAB        /STOP OTHER COAX USER
```

```
6650  7350     CLA CMA CLL RAR        /SET A LONG DELAY
6651  4775     JMS I DELAYX  /WAIT FOR SPECTRGRAPH CONTROL TO FINISH
6652  1055     TAD ARG6
6653  0374     AND P7
6654  1365     TAD LIST
6655  3031     DCA TEMPS0
6656  1431     TAD I TEMPS0
6657  3014     DCA UNIT
6660  2013  TEST, ISZ BYPASS /SET ERROR PRINT BYPASS
6661  2015     ISZ COUNT
6662  7410     SKP
6663  5266     JMP GOSYNC
6664  4200  INIT, JMS XMIT    /PUTS MPX IN PHASE, WITH NO ERROR PRINT
6665  5260     JMP TEST
6666  3013  GOSYNC, DCA BYPASS        /CLEAR ERROR PRINT BYPASS
6667  1057  @  TAD ARG8
6670  7650     SNA CLA
6671  4776     JMS I  SYNCX   /WAIT FOR 4K SCANNER MEMORY CYCLE
6672  1017     TAD FUNCP
6673  0363     AND P5777      /MEM. CLOCK OFF
6674  6453     FUNLOD         /STOP 4K MEM.
6675  7200     CLA
6676  1017     TAD FUNCP
6677  4764     JMS I SFUNLX
6700  1372     TAD M1000
6701  3015     DCA COUNT
6702  7332     CLA STL RTR    /CORE BUFFER 1, ADDRESS 2000
6703  3061     DCA DATA
670H  1244     TAD SETUP
6705  1377     TAD CENTES     /CHECK FOR CENTERING ONLY
6706  7650     SNA CLA
6707  5342     JMP GETOUT     /YES
6710  1373  NEXT, TAD M10
6711  3060     DCA COUNT8
6712  1054     TAD ARG5
6713  7650     SNA CLA
6714  5327     JMP PUTING     /SETTING SWEEPS(SEND)
6715  6454  STEPRD, MCSTEP
6716  2060     ISZ COUNT8
6717  5315     JMP STEPRD
6720  1373  GET, TAD M10      /CHANGE SEND TO RECEIVE
6721  1367     TAD PORT7
6722  4200     JMS XMIT
6723  5320     JMP GET        /ERROR
6724  6353     MUX DAREAD     /READ 12 BIT WORD
6725  3461     DCA I DATA
6726  5337     JMP ONGO
6727  1461  PUTING, TAD I DATA
6730  7421     MQL
6731  1366     TAD PORT18
6732  4200     JMS XMIT
6733  5327     JMP PUTING     /ERROR
6734  6454  STEPW, MCSTEP
6735  2060     ISZ COUNT8
6736  5334     JMP STEPW      /8 PULSES PER STEP
6737  2061  ONGO, ISZ DATA
6740  2015     ISZ COUNT
6741  5310     JMP NEXT
6742  1075  GETOUT, TAD NORMAL
6743  4764     JMS I SFUNLX
```

```
6744  1053  SETIT, TAD ARG4
6745  7421    MQL
6746  1366    TAD PORT18      /SETTING 'CENTER' FOR SWEEPS
6747  4200    JMS XMIT
6750  5344    JMP SETIT       /ERROR
6751  1370    TAD CABL18
6752  7421    MQL
6753  7330    CLA STL RAR     /4000 FOR PULSE1
6754  4200    JMS XMIT
6755  5344    JMP SETIT       /ERROR
6756  1075    TAD NORMAL
6757  6453    FUNLOD
6760  7300    CLA CLL
6761  6356    MUX LAMOFF
6762  5644    JMP I SETUP
              /
6763  5777  P5777, 5777
6764  6546  SFUNLX, SFUNL
6765  6505  LIST, UNIT0
6766  0400  PORT18, 400        /CABLE FOR DATA
6767  7000  PORT7, 7000
6770  0020  CABL18, 20         /CABLE FOR PULSES
6771  0040  CABLE7, 40
6772  7000  M1000, -1000
6773  7770  M10, -10
6774  0007  P7, 7
6775  6561  DELAYX, DELAY
6776  6566  SYNCX, SYNC
              /
              /
6777  1274  CENTES, -CENGO
```

```
                /
                /XMIT
                /STANDARD SUBROUTINES TO RUN SERIAL MULTIPLEXER
                /
                /CALL:TAD N      /N+1 TRIES BEFORE ERROR PRINT
                /    JMS XMIT
                /    FUNCTION CODE
                /    'DATA'
                /    ERROR RETURN
                /    NORMAL RETURN
                /
                MUX=6350
                OKSKIP=1
                DUNSKP=2
                DAREAD=3
                FREAD=4
                FLOAD=5
                DLOAD=7
                /
                *XMITX
0100    6042    XMIT
                *6042
6042    0000    XMIT,  0
```

Continued on next page

```
P2
6043  7040    CMA
6044  3316    DCA BYPASS     /SET ERROR BYPASS COUNT
6045  1642  ENTRY, TAD I XMIT        /GET FUNCTION
6046  2242    ISZ XMIT
6047  6355    MUX FLOAD      /SET UP FUNCTION
6050  3310    DCA FUNC
6051  6354    MUX FREAD      /NOW CHECK IT
6052  7041    CIA
6053  1310    TAD FUNC
6054  7640    SZA CLA
6055  5276    JMP FAULT      /FUNCTION DIDN'T LOAD
6056  1642  GO, TAD I XMIT   /GET 'DATA'
6057  6357    MUX DLOAD      /LOAD AND TRANSMITS
6060  7344    CLL CLA CMA RAL
6061  7104    CLL RAL        /SET -4
6062  3110    DCA TEMP
6063  2110    ISZ TEMP
6064  5263    JMP .-1        /WAIT 20 MICROSEC
6065  6352    MUX DUNSKP     /CHECK IF RECEIVER RESPONDED
6066  5276    JMP FAULT      /NO RESPONSE
6067  6354    MUX FREAD
6070  7041    CIA
6071  1310    TAD FUNC       /COMPARE SENT AND ECHOED FUNCTION
6072  6351    MUX OKSKIP
6073  7240    CLA CMA        /PARITY ERROR
6074  7650    SNA CLA
6075  5305    JMP OK
6076  2316  FAULT, ISZ BYPASS        /PRINT IF BYPASS = 7777
6077  5312    JMP SYNTES
6100  4422    JMS I MESAGX
6101  1525    TEXT /MU
6102  3077  X?
6103  0000  /
6104  7410    SKP    /ERROR RETURN
6105  2242  OK, ISZ XMIT
6106  2242    ISZ XMIT
6107  5642    JMP I XMIT
            /
6110  0000  FUNC, 0
6111  0000  TEMP, 0
            /
6112  7240  SYNTES, CLA CMA
6113  1242    TAD XMIT
6114  3242    DCA XMIT       /REPEAT
6115  5245    JMP ENTRY
            /
6116  0000  BYPASS, 0
6117  0000  TEMP2, 0

.
```

MULTIPLEXER DIAGNOSTIC

A general purpose focal instruction to obtain data or send data through the Serial Data Converter and Multiplexer (EL-331, EL-505) is available. It cannot both send data and receive data with one transmission (although the electronics are capable of this). This command is of the form :



If an error occurs, the program will type out an error message and return to focal with a negative number in D. There are three types of errors that may occur. If the operator improperly codes the data for the arguments of FMPX, the program will respond with "FORMAT ERROR!" If the coax cables are in use by another device (i.e. some other device that uses the coax cables was not turned off by the "look at me" pulse) the program will respond with "LINE NOT FREE". If an error occurs in the transmission the program responds with "TRANSMISSION ERROR", followed by two rows of binary numbers representing the Function and Data words which were transmitted and received.

This is an X NAME () command and can be obtained from T. Ricketts.

T. RICHETTS
TAPE 2, FILE 3
TAPE 3, FILE 4
IBM ASSEMB. #2
X NAME (3)
2-11-76
SCAN 23-F #1
X NAME (5)
2-2-76

```
 PALP
*OUT-S:XMPX
*
*IN-S:CON0,S:MPX1,S:MPX2,S:MPX3
*
*
*
*
*OPT-T

US COMP+0000

US CTEM+0000

ARG1    0050
ARG10   0061
ARG10H  0017

                /CON0
                XLIST           3 4 5 6 7 8 9
                PAUSE   /X MPX(U,D,S,L,P,N,B)
                        /FOCAL COMMAND FOR THE SERIAL MULTIPLEXER
                        /U=UNIT; 0=NONE,#1-#7
                        /D=DEVICE; 0=NONE, #1-14=INPUT,#15-18=OUTPUT
                        /S=OUTPUT DATA (0-4096)
                        /L=LEVEL; 0=NONE;1=LEVEL 1;2=LEVEL 2; +=SET TO 1; -=SET TO 0
                        /P=PULSE; 0=NONE,1=PULSE 1, 2=PULSE 2
                        /N=NUMBER OF RETRYS BEFORE ABORTION & PRINTOUT
                        /IF N IS NEGATIVE DELETE PRINTOUTS
                        /B=TYPE BELL ONLY FOR ERRORS
                        /
                *FNKB1+66
0732    1230    1230    /MPX
                        /
                *KB1+66
0226    6200    XMPX
                        /
                *6200
6200    0000    XMPX,   0000
6201    7300            CLA CLL
6202    1015            TAD ARG8H       /NO PRINTOUTS?
6203    7510            SPA
6204    2061            ISZ ARG10       /NO PRINTOUTS!
6205    3370            DCA TEM
6206    1057            TAD ARG8        /# OF RETRYS
6207    4754            JMS I XCOMP     /RESTORE ARG8 FOR NEG INPUT
6210    3373            DCA XCNT        /SET UP COUNTERS FOR # OF RETRYS
6211    1373            TAD XCNT
6212    3771            DCA I COUNT
6213    1060            TAD ARG9        /PRINT BELL ONLY?
6214    7640            SZA CLA
6215    2017            ISZ ARG10H      /YES
6216    1052            TAD ARG3        /UNIT #
6217    1364            TAD CM7         /CHECK FORMAT
6220    7740            SMA SZA CLA
6221    5775            JMP I XFORM     /UNIT # TOO LARGE
6222    1052            TAD ARG3
6223    4756            JMS I XSWTCH
6224    7012            RTR
6225    7012            RTR
6226    7010            RAR
6227    3052            DCA ARG3
```

```
6230  1054          TAD ARG5        /OUTPUT DATA
6231  3372          DCA OUTPUT
6232  1013          TAD ARG6H
6233  3370          DCA TEM
6234  1055          TAD ARG6        /LEVEL?
6235  4754          JMS I XCOMP     /RESTORE ARG6 FOR NEG INPUT
6236  1056          TAD ARG7        /PULSE?
6237  7440          SZA
6240  5757          JMP I XPSLV     /INSERT PULSE OR LEVEL
6241  1366          TAD C14         /DEVICE # >14?
6242  7041          CIA
6243  1053          TAD ARG4
6244  7540          SZA SMA
6245  5254          JMP TST         /YES; OUTPUT DATA
6246  7300          CLA CLL
6247  1053          TAD ARG4        /NO; INPUT DEVICE CODE
6250  4756          JMS I XSWTCH
6251  7010          RAR
6252  1365          TAD C200        /INPUT CODE
6253  5267          JMP GO+3
6254  3370  TST,    DCA TEM         /TEST FORMAT
6255  1370          TAD TEM
6256  1363          TAD CM4
6257  7740          SMA SZA CLA
6260  5775          JMP I XFORM     /DEVICE # TOO LARGE
6261  1363          TAD CM4
6262  7041          CIA
6263  1370          TAD TEM
6264  4756  GO,     JMS I XSWTCH
6265  7010          RAR
6266  1367          TAD C400        /OUTPUT CODE
6267  1052          TAD ARG3        /COMBINE UNIT,DEVICE, & DIRECTION
6270  7012          RTR
6271  7012          RTR
6272  3053          DCA ARG4        /PROPER FORMAT FOR FUNCTION
6273  1055          TAD ARG6        /SEND A LEVEL?
6274  7650          SNA CLA
6275  5304          JMP LAM         /NO
6276  1013          TAD ARG6H       /YES;SEND A 1 OR A 0?
6277  7500          SMA
6300  7001          IAC             /SET TO 1
6301  7001          IAC             /SET TO 0
6302  1053          TAD ARG4
6303  3053          DCA ARG4
6304  6353  LAM,    6353            /LOOK AT ME; SPECT. OFF
6305  6002          IOF
6306  4761          JMS I WAIT      /WAIT 150 MICROSEC.
6307  1053  XMIT,   TAD ARG4
6310  6355          6355            /LOAD FUNCTION REGISTER
6311  7300          CLA CLL
6312  4762          JMS I DLAY      /WAIT 15 MICRO SEC.
6313  6354          6354            /READ FUNCT.
6314  7041          CIA
6315  1053          TAD ARG4        /SAME AS LOADED?
6316  7650          SNA CLA
6317  5323          JMP .+4         /YES; GO ON TO TRANSMIT
6320  2373          ISZ XCNT        /NO; TRY AGAIN N TIMES
6321  5304          JMP LAM
6322  5776          JMP I LINE      /EXIT; LINE NOT FREE
6323  1372          TAD OUTPUT
```

```
6324  6357            6357              /LOAD DATA & TRANSMIT
6325  7300            CLA CLL
6326  6352  FLAG,     6352              /SKIP, TRANSMISSION DONE
6327  5755            JMP I XRSTRT      /AUTO RESTART IF NO DATA RECEIVED
6330  6351            6351              /SKIP, NO ERRORS
6331  5760            JMP I XMER        /TRANSMISSION ERROR
6332  6354            6354              /READ FUNCTION
6333  3370            DCA TEM
6334  1053            TAD ARG4          /XMITTED FUNCT.
6335  7041            CIA
6336  1370            TAD TEM           /RECEIVED FUNCT.
6337  7440            SZA               /SAME?
6340  5760            JMP I XMER        /NO; ERROR
6341  6353            6353              /YES; READ DATA
6342  7040            CMA
6343  3051            DCA ARG2
6344  6356            6356              /LOOK AWAY; SPECT. ON
6345  5600  RETURN,   JMP I XMPX
                      /
6346  1374  BELL,     TAD K207
6347  4527            JMS I TYPEX       /TYPE A BELL
6350  7330  EXIT,     7330              /SET AC TO 4000
6351  3050            DCA ARG1          /MAKE DATA NEGATIVE
6352  6356            6356              /LOOK AWAY; SPECT. ON
6353  5600            JMP I XMPX
                      /
6354  6473  XCOMP,    COMP
6355  6400  XRSTRT,   RESTRT
6356  6440  XSWTCH,   SWITCH
6357  6403  XPSLV,    PSLV
6360  6057  XMER,     ERROR
6361  6465  WAIT,     DELAY
6362  6117  DLAY,     MPXWT
6363  7774  CM4,      7774
6364  7771  CM7,      7771
6365  0200  C200,     0200
6366  0016  C14,      0016
6367  0400  C400,     0400
6370  0000  TEM,      0000
6371  6154  COUNT,    CNT
6372  0000  OUTPUT,   0000
6373  0000  XCNT,     0000
6374  0207  K207,     0207
6375  6044  XFORM,    FORMAT
6376  6125  LINE,     LINTFR
                     /
                      /
                      /MPX2
                      /
                      /
                *6400
6400  2320  RESTRT,   ISZ TEM4          /DELAY FOR AUTO RESTART
6401  5717            JMP I XFLAG
6402  5721            JMP I XXERR
                      /
6403  3307  PSLV,     DCA TEM1
6404  1307            TAD TEM1
6405  7041            CIA               />2?
6406  7001            IAC
6407  7001            IAC
```

```
6410  7710          SPA CLA
6411  5722          JMP I XXFORM     /YES; FORMAT ERROR
6412  1053          TAD ARG4         /DEVICE
6413  7041          CIA
6414  1315          TAD K12          />12?
6415  7710          SPA CLA
6416  5722          JMP I XXFORM     /FORMAT ERROR
6417  7240          CLA CMA
6420  1053          TAD ARG4         /NO; GET DEVICE CODE-1
6421  3225          DCA SHIFT
6422  7421          7421             /MQL; CLEAR MQ REG.
6423  7001          IAC
6424  7417          7417             /LSR; SHIFT A 1 TO THE RIGHT BIT
6425  0000  SHIFT,  0000
6426  7300          CLA CLL
6427  7501          7501             /MQA; CODE FOR OUTPUT ROUTING
6430  3711          DCA I OUT
6431  1055          TAD ARG6         /LEVEL?
6432  7650          SNA CLA
6433  5236          JMP .+3          /NO; MUST BE PULSE
6434  7001          IAC
6435  7001          IAC
6436  1307          TAD TEM1
6437  5716          JMP I XGO
                    /
6440  0000  SWITCH, 0000
6441  3307          DCA TEM1
6442  3310          DCA TEM2
6443  3312          DCA TEM3
6444  1313          TAD KM4          /# OF BITS TO SWITCH
6445  3320          DCA TEM4
6446  7300          CLA CLL
6447  1307          TAD TEM1
6450  7010          RAR
6451  3307          DCA TEM1         /SHIFT AC11 INTO LINK,SAVE REST
6452  1310          TAD TEM2         /SWITCHED WORD BUFFER
6453  7430          SZL              /IS LINK 0?
6454  2312          ISZ TEM3         /NO; PUT IN A 1
6455  7104          CLL RAL          /SHIFT TEM2 LFT FOR EACH
6456  1312          TAD TEM3         /RT SHIFT OF TEM1
6457  3310          DCA TEM2
6460  3312          DCA TEM3
6461  2320          ISZ TEM4         /LAST SHIFT?
6462  5247          JMP SWITCH+7     /NO
6463  1310          TAD TEM2         /YES
6464  5640          JMP I SWITCH
                    /
6465  0000  DELAY,  0000
6466  1314          TAD KM30
6467  3312          DCA TEM3
6470  2312          ISZ TEM3         /WAIT 150 MICRO SEC.
6471  5270          JMP .-1
6472  5665          JMP I DELAY
                    /
6473  0000  COMP,   0000
6474  3305          DCA CTEM
6475  1706          TAD I XTEM
6476  7640          SZA CLA
6477  5302          JMP .+3          /NUMBER IS NEG
6500  1305          TAD CTEM         /NUMBER IS POS.
```

```
6501  5673          JMP I COMP
6502  1305          TAD CTEM
6503  7041          CIA
6504  5673          JMP I COMP
6505  0000  CTEM,    0000
6506  6370  XTEM,    TEM
                    /
6507  0000  TEM1,    0000
6510  0000  TEM2,    0000
6511  6372  OUT,     OUTPUT
6512  0000  TEM3,    0000
6513  7774  KM4,     7774
6514  7740  KM30,    7740
6515  0014  K12,     0014
6516  6264  XGO,     GO
6517  6326  XFLAG,   FLAG
6520  0000  TEM4,    0000
6521  6057  XXERR,   ERROR
6522  6044  XXFORM,  FORMAT
                    /
                    /
                   /MPX3
                    /
                    /FOCAL COMMAND FOR SERIAL MULTIPLEXER
                    /
            *6044
6044  4340  FORMAT,  JMS CHECK          /CHECK FOR PRINTOUT
6045  4422           JMS I MESAGX
6046  0617           TEXT /FO
6047  2215  RM
6050  0124  AT
6051  4005   E
6052  2222  RR
6053  1722  OR
6054  4041   !
6055  0000  /
6056  5751           JMP I EXXIT
                     /
6057  2354  ERROR,   ISZ CNT            /TRY N TIMES
6060  5752           JMP I XXMIT
6061  4340           JMS CHECK
6062  4422           JMS I MESAGX
6063  2422           TEXT /TR
6064  0116  AN
6065  2315  SM
6066  1123  IS
6067  2311  SI
6070  1716  ON
6071  4005   E
6072  2222  RR
6073  1722  OR
6074  4041   !
6075  0000  /
6076  7300           CLA CLL
6077  3126           DCA INTRUP          /KEEP INTURRUPT OFF
6100  4530           JMS I CRLFX         /TYPE CRLF
6101  7300           CLA CLL
6102  1053           TAD ARG4            /FUNCT. XMITTED
6103  4757           JMS I XERTYP        /TYPE IT
6104  4777           JMS SPACE
```

```
6105  1776          TAD OUTPUT      /DATA XMITTED
6106  4757          JMS I XERTYP
6107  4530          JMS I CRLFX
6110  6354          6354            /FUNCT. RECEIVED
6111  4757          JMS I XERTYP
6112  4777          JMS SPACE
6113  6353          6353            /DATA RECEIVED
6114  4757          JMS I XERTYP
6115  4530          JMS I CRLFX
6116  5751          JMP I EXXIT
                    /
6117  0000  MPXWT,   0000
6120  1350          TAD XM4
6121  3353          DCA MARK        /WAIT 15 MICRO SEC.
6122  2353          ISZ MARK
6123  5322          JMP .-1
6124  5717          JMP I MPXWT
                    /
6125  4340  LINTFR,  JMS CHECK      /LINE NOT FREE
6126  4422          JMS I MESAGX
6127  1411          TEXT /LI
6130  1605  NE
6131  4016   N
6132  1724  OT
6133  4006   F
6134  2205  RE
6135  0541  E!
6136  0000  /
6137  5751          JMP I EXXIT
                    /
6140  0000  CHECK,   0000
6141  1061          TAD ARG10       /NO PRINTOUT?
6142  7640          SZA CLA
6143  5751          JMP I EXXIT     /YES
6144  1017          TAD ARG10H      /TYPE BELL?
6145  7640          SZA CLA
6146  5756          JMP I XBELL     /YES
6147  5740          JMP I CHECK
                    /
6150  7774  XM4,    7774
6151  6350  EXXIT,  EXIT
6152  6304  XXMIT,  LAM
6153  0000  MARK,   0000
6154  0000  CNT,    0000
6155  0000  MPXCNT, 0000
6156  6346  XBELL,  BELL
6157  6600  XERTYP, ERRTYP
                    /
6176  6372
6177  6617
                  *6600
6600  0000  ERRTYP, 0000
6601  3231          DCA TBUF
6602  1234          TAD KM12
6603  3232          DCA TCNT        /TYPE 12 BITS
6604  1231  XTYPE,  TAD TBUF
6605  7004          RAL             /PUT MSB IN LINK
6606  3231          DCA TBUF
6607  7630          SZL CLA
6610  7001          IAC             /TYPE A 1 IF LINK IS 1
```

```
6611  1233          TAD K260          /ASCII CODE
6612  4527          JMS I TYPEX       /TYPE BIT
6613  2232          ISZ TCNT          /12 BITS?
6614  5204          JMP XTYPE         /NO
6615  7300          CLA CLL           /YES
6616  5600          JMP I ERRTYP
                    /
6617  0000  SPACE, 0000
6620  7300          CLA CLL           /TYPE 5 SPACES
6621  1235          TAD KM5
6622  3232          DCA TCNT
6623  1236          TAD BSPACE
6624  4527          JMS I TYPEX
6625  2232          ISZ TCNT
6626  5223          JMP .-3
6627  7300          CLA CLL
6630  5617          JMP I SPACE
                    /
6631  0000  TBUF,  0000
6632  0000  TCNT,  0000
6633  0260  K260,  0260
6634  7764  KM12,  7764
6635  7773  KM5,   7773
6636  0240  BSPACE, 0240
                    /

.
```

Maintenance

This section is primarilly intended for maintenance personnel who will need some information on the electronics circuits contained in the multiplexer.

Section 1 General Operation

Refer to drawing EL-331-3L for the overall block diagram of the Serial Data Converter and Multiplexer, and to drawing EL-505-2L for the block diagram of the Computer Interface.

A. Computer Interface

Data from the computer is sent via a 12 wire bus (BACO-11) which is routed to the inputs of the function (B23,24) and Data (A23,24) shift registers. Data from these registers is gated back to the 12 bit Accumulator input bus by two Data strobe cards (EL-195). These gates and registers are controlled by IOT pulses generated by the Device Selector (B19,20). Control of the shifting of data from the registers to the output coax and back is accomplished by the shift control card (EL-326). This card is the heart of the whole multiplixer system and will be discussed in greater detail later.

The data and shift lines from the shift control card are split into two separate coax busses by the driver card (A19,20). One bus drives the tub while the other drives the R.O. room and coudé room. This was done so that the coaxes would not have to make a return trip from the telescope to drive another system.

The actual driving of the 50-ohm coax lines is done by the 50 ohm cable driver cards (EL-339). These cards are separated into two sections, a driver and a cable terminator. The termination is only added to the cards at the two extreme ends of the cable. Data levels on the lines

are 0 and +3 volts, with 0 volts being a "1" and +3 volts being a "0".

B. Serial Multiplexer

Up to 7 serial multiplexer boxes, each with its own identification number, can be driven from the computer interface. The central portion of the Serial Multiplexer is similar to the same section of the computer interface. It consists of the Cable Driver, a Shift Control Card, and two Shift Registers. The Shift Control Card is connected so that it will automatically respond to a transmission from the computer.

On the left side of the drawing (EL-331-3L) is a series of 'Data Strobe" cards. Each is connected to one of 14 input cables. The outputs are "or"ed and bussed to the input of the Data Register (B11,12). The output bus from the Data register drives a series of cards on the right of the drawing. The four buffers on the bottom (B21-B28) store the data and drive four output connectors. The four cards at the top right produce the levels and pulses which are sent to the input connectors via the data strobes.

The selection of which of these cards is to send or receive data is accomplished by the "16 DIGIT Decoders" (B3,4; B5,6). The "function" word is stored in a buffer if the multiplexer unit is selected. The buffer output is coupled back to the Function Shift Register for transmission back to the computer. It is also coupled to the two decoders. One decoder selects the proper input data strobe card, while the other selects the output source for the data.

To produce pulses on the 17 pin cables the "load data" pulse from the shift control card is delayed 10 μsec by a dual one shot card (A31) to produce a "clear" pulse for the two shift Register cards (top right) which produce the pulses. The "load data" pulse strobes the decoder to produce a pulse on the output. If device 1 or 2 has been decoded

and "output data" selected, then a load pulse will cause the data to be loaded into one of the Shift Registers. The corresponding outputs will go high sending a "one" to the input cables selected by the data word. Then 10μsec later the "clear" pulse will clear the registers, removing the "one" from the selected cables. The shift register cards are used only as buffers, not as shift registers. We used these cards instead of simple flip flops because there was no clearable buffer available when the system was designed.

The 'levels' to the 17 pin cables are produced in much the same way as the pulses. If device 3 or 4 is selected as an output one of the two MPX. buffers (EL-465) will be selected. The data lines then select which levels in the buffer will be changed. Two lines from the function buffer determine whether the change will be to a "1" or to a "0".

Section 2 Operation of Specific Cards

A. Shift Control EL-326

Note that this circuit is used in both computer interface and multiplex boxes. Refer to drawings EL-326-3L and EL-326-3D. When a transmission is desired, a "transmit" pulse is applied to pin 2D. This pulse sets the "Transmit-Receive" flip flop (G9) into the "transmit" state. It also clears the flag (also G9), and the divide by 30 counter (G10, G11).

The "Transmit" level from G9 pin 15 enables G5 at pin 4 to allow the data to pass from the shift registers to the drivers. It also enables the 5 MHz oscillator (G1) (after a 1μsec delay that permits the "Transmit" pulse to finish clearing the counter). The "Transmit" level is fed to G1 pin 13 after the delay and causes G1 pin 12 to go

low, closing gate G8 to prevent the shift pulses on the coax from being counted. The low on G1 pin 12 will cause G1 pin 10 to go high enabling the oscillator to run. The oscillator consists of three inverters connected in a loop. The frequency is determined by the delay time in the loop. This time is adjusted by means of the capacitor between pins 3 and 4. The value of the capacitor is chosen to give about 5 MHz at the output.

The pulses from the oscillator are coupled through G2 to pin 2E to provide the shift pulses which are sent via the coax to the shift control card in the multiplexer being serviced. The pulses are also routed through G8 (pin 1) to provide shift pulses (pin 1P) for the shift registers, and to clock the ÷ 30 counter (G10, G11). G10 divides the pulses by 10 while the ÷ 5 section of G11 is used to further divide them by 3. G8 (Pins 9, 10) decode the count of 30. Pin 8 will go low when the number of pulses exceeds 30. This signal closes the gate (G5) for the shift pulses, and enables the decade decoder (G12) to decode the succeeding pulses.

After these 30 pulses, the data should be present in the two shift registers in the receiver. Two checks are made of that data. Pins 1D through 1H will pass the unit code selected to a decade decoder (G13). If the code matches the location of the jumper, G2 pin 10 (pin 2K) will go high indicating that the Multiplexer box has been selected. Also, if the two error check inputs (2H, 2J) are both low (no errors in the shift register), G4 pin 4 (Pin 2¥) will go high. This signal gates an IOT pulse to produce a skip for the computer if there is no error. It also produces a low at the output of G7 (Pin 11) if the unit has been selected. This signal is inverted and routed to G7 pin 10 where

it is gated at shift pulse "31" time to clock the ÷ 2 section of G11. This flip flop, when set, indicates that the proper code has been received to select the unit and that no errors are present. It will enable the succeeding shift pulses to load various registers within the multiplexer.

Shift pulse "32" is gated through G4 (pin 11) to load the function register into the buffer. Pulse "33" loads the Data into the output buffers. Pulse "34" clears the shift registers, and pulse "35" loads them from the input gates and the function buffer. The output of G12 pin 6, which loads the shift registers, goes low at pulse "35" time and back high at pulse "36". This signal is inverted to produce Reset (Pin 2S) which is used to clock the flip-flops (G9). This will switch the transmit-receive flip-flop in the transmitter to the receive state, turning off the oscillator. The flip-flop in the receiver is switched to the transmit state and will begin to transmit back to the computer. The 36th pulse causes G12 pin 7 to go low. This signal is delayed through G3 and G8 to produce Reset' (pin 2u) which resets the ÷ 30 counter.

The timing sequence will repeat now with the computer acting as the receiver. At the end of the second transmission the Reset pulse will clock G9 pin 6 to set the Flag (Pin 2 P). This signal will tell the computer that the data has been received from the multiplexer. The flag is gated with an IOT pulse in G7 to produce a skip for the computer.

B. Shift Register EL-324

This card consists of three 5 bit shift registers (G1, G2, G3). Each chip contains 4 bits of data plus one error check bit. The clear

and preset inputs are straight forward, driving all chips in parallel. It should be noted that the preset can only set ones into the register, it cannot change a one to a zero. Therefore the registers are normally cleared before being preset.

The preset inputs arrive on pins D1-S1, D1 being most significant. The "A" inputs of the three chips are hard wired to encode a 1, 0, and 1 into G1, 2, and 3. These bits are used to detect transmission errors. The shift clock (pin U2) causes the data to shift serially from G1 to G3. Serial data arrives at pin T2 and leaves at pin S2. The contents may also be read in parallel at D2-S2.

The error code is gated by G4 to produce $\overline{\text{Error Check}}$ at pin V1. If the check bits are 1, 0, and 1 the $\overline{\text{Error check}}$ signal will go low. This signal will normally change as data is shifted through the shift register, but it is only checked after the shift clock has been disabled and data is stable.

C. Data Strobe (Cable) EL-329

This card is similar to the EL-195 with the exception that the input arrives on a cable at the rear of the card rather than from pins D1-S1. There are also four additional lines which couple the cable to pins. These are used to feed the levels and pulses back to the input connector.

D. 16 DIGIT Decoder EL-330

This card contains one IC which accepts a 4 bit binary code (pins D_1-H_1) and decodes it into 16 outputs. There are two additional gates (G1 and G2) which must be low to enable the outputs. These gate inputs are used in the multiplexer by the read and write commands in the function word, and by the load data pulse.

E. 50 Ohm Cable Driver EL-339

This circuit consists of two parts, a cable driver and a terminator. The terminator is only used on the two extreme ends of the coax. It consists of two transistors acting as a darlington emmitter follower and will produce about 3V at pin R. From pin R a 50 ohm resister connects to the coax providing both a termination and a pull-up for the driver.

The driver consists of two transisters. The first is an emitter follower while the second is an open collector switch. The coax cables are connected directly to the collector of the second transister to minimize reflections.

F. 12 Bit Buffer, Selectable Bit Change EL-465

This card consists of 12 J-K flip-flops (G2-G7) and the select gates (G1).

The inputs (D1-S1) are applied to the clock inputs of the respective flip-flops. The "J" and "K" inputs come from the "load data" pulse which has been gated by the "set to 1" for "J", and the "set to 0" for "K".

The shift registers are cleared to 0 immediately after the end of the "load data" pulse. This will cause the clock inputs of those flip-flops which have been selected, via the data word, to make a transition from 1 to 0, clocking the "J" or "K" input (whichever is selected) into the flip-flop. The resister and capacitor at the output of each gate cause the falling of the "J" and "K" inputs to be delayed until after the clock lines have gone to 0.

Section 3 Troubleshooting Hints

Most troubles will be due to improper cable lengths, no termination, or no power to the selected multiplexer unit. Check those first.

If trouble should develop, use a program which will continually transmit a fixed data pattern. The FMPX command is good for trouble-shooting. It may be set to repeat any number of times and can provide some error diognostics to narrow down the area of trouble.

Using the FMPX command, ask for data from a connector that is open. A statement such as TYPE FMPX (1,1) will read back data from cable 1, Multiplexer unit 1. If the connector is open the result should be 4095. If something else is typed something may be wrong with that connector or its data strobe. If several connectors are wrong but no error diagnostics are typed out, then a scope will be needed to check out the multiplexer. The computer may respond with two lines of numbers such as:

```
100010001000    000000000000
000000000000    000000000000
```

If this happens it indicates that the computer is getting no response from any multiplexer. Check that power is connected to the selected Multiplexer box. The first line is the function and data words in binary form which were transmitted while the second line is the received words. The transmitted function word may be checked to verify that it is correct.

The next step is to use a scope and examine the shift and data lines at the Multiplexer box. The best place to do this is on the Shift Control Card, EL-326 in location A, B-32. The shift pulses are on A32-V. Verify that there are 36 (or 37) pulses. Check to see if the multiplexer is responding with a second group of 36 pulses. Pin

A32-N "<30" is a good sync point to check on status of signals that should be stable only after pulse 30. Check pin B32-R for Load enable. This signal will be present only if the particular Multiplexer box was selected and the error code was correct. If not present, then check B32-K for unit select and B32-V for error code; work back from each of these to the source. If Load Enable is present, check $\overline{\text{Reset}}$ (B32-S) and Reset' (B32-U) for proper timing. If still correct check the transmit line (A32-L) to see if the multiplexer is being told to transmit. If so the oscillator must be at fault; check B32-E for the shift pulses out.

In one of the above stages something will be missing if the unit is not responding. If however the unit is transmitting properly but the data is wrong the various support cards should be checked.

APPENDIX A

CONNECTION DIAGRAM OF MULTIPLEXER SYSTEM



APPENDIX B

CONNECTOR WIRING

INPUT CONNECTOR

PIN	SIGNAL		
A	AC_o	MSB	
B	AC_1		
C	AC_2		
D	AC_3		
E	AC_4		
F	AC_5	Data Input	
G	AC_6	"1" = 0 Volts	
H	AC_7	"0" =+4 Volts	
J	AC_8		
K	AC_9		
L	AC_{10}		
M	AC_{11}	LSB	
N	Pulse (1)	Pulse Width = 10μs	
P	Pulse (2)	Control out "1" = +4 Volts	
R	Level (3)		
S	Level (4)	"0" = 0 Volts	
T	Commond Ground		

OUTPUT CONNECTOR

PIN	SIGNAL	
A	BAC_o	MSB
B	BAC_1	
C	BAC_2	
D	BAC_3	
E	BAC_4	
F	BAC_5	Data Output
G	BAC_6	"1" = +4 Volts
H	BAC_7	"0" = 0 Volts
J	BAC_8	
K	BAC_9	
L	BAC_{10}	
M	BAC_{11}	LSB
N	Pulse (1)	Paralled with Input Connectors J5-J8
P	Pulse (2)	
R	No Connection	
S	No Connection	
T	Common Ground	

APPENDIX C

FUNCTION WORD STRUCTURE

FUNCTION WORD

ACCUMULATOR BIT		SIGNAL	
MSB	0	1	Cable # See Below for Decoding, (P. 35)
	1	2	
	2	4	
	3	8	
	4	1	Unit # (Crate)
	5	2	
	6	4	
	7	Send Data Enable	
	8	Receive Data Enable	
	9	Spare	
	10	Set Level = "1"	
LSB	11	Set Level = "0"	

Note reversal of most and least significant bits

APPENDIX C

Continued

Cable # Codes

Code	Receive Data	Send Data
0	No Connection	No Connection
1	J1	Pulse (1)
2	J2	Pulse (2)
3	J3	Level (3)
4	J4	Level (4)
5	J5	J15
6	J6	J16
7	J7	J17
8	J8	J18
9	J9	No Connection
10	J10	No Connection
11	J11	No Connection
12	J12	No Connection
13	J13	No Connection
14	J14	No Connection
15	No Connection	No Connection

APPENDIX D

IOT ALLOCATION

6351	Skip if no transmission error
6352	Skip on "transmit done" flag
6353	Read data register into accumulator
6354	"Look at me"; read function register into accumulator
6355	Clear and load function register from accumulator
6356	"Look away"
6357	Clear and load data register from accumulator, and start transmission